pragma solidity ^0.4.16;

interface tokenRecipient { function receiveApproval(address _from, uint256 _value, address _token, bytes _extraData) external; }

contract TokenERC20 {
    // Public variables of the token
    string public name = "SurveyToken";
    string public symbol = "SRT";
    uint8 public decimals = 18;
    uint256 public totalSupply;

    // This creates an array with all balances
    mapping (address => uint256) public balanceOf;
    mapping (address => mapping (address => uint256)) public allowance;

    // This generates a public event on the blockchain that will notify clients
    event Transfer(address indexed from, address indexed to, uint256 value);

    // This notifies clients about the amount burnt
    event Burn(address indexed from, uint256 value);

    /**
     * Constrctor function
     *
     * Initializes contract with initial supply tokens to the creator of the contract
     */
    function TokenERC20(uint256 initialSupply) public {
        totalSupply = initialSupply * 10 ** uint256(decimals);  // Update total supply with the decimal amount
        balanceOf[msg.sender] = totalSupply;
    }

    /**
     * Internal transfer, only can be called by this contract
     */
    function _transfer(address _from, address _to, uint _value) internal {
        // Prevent transfer to 0x0 address. Use burn() instead
        require(_to != 0x0);
        // Check if the sender has enough
        require(balanceOf[_from] >= _value);
        // Check for overflows
        require(balanceOf[_to] + _value > balanceOf[_to]);
        // Save this for an assertion in the future
        uint previousBalances = balanceOf[_from] + balanceOf[_to];
        // Subtract from the sender
        balanceOf[_from] -= _value;
        // Add the same to the recipient
        balanceOf[_to] += _value;
        Transfer(_from, _to, _value);
        // Asserts are used to use static analysis to find bugs in your code. They should never fail
        assert(balanceOf[_from] + balanceOf[_to] == previousBalances);
    }

    /**
     * Transfer tokens
     *
     * Send `_value` tokens to `_to` from your account
     *
     * @param _to The address of the recipient
     * @param _value the amount to send
     */
    function transfer(address _to, uint256 _value) public {
        _transfer(msg.sender, _to, _value);
    }

    /**
     * Transfer tokens from other address
     *
     * Send `_value` tokens to `_to` in behalf of `_from`
     *
     * @param _from The address of the sender
     * @param _to The address of the recipient
     * @param _value the amount to send
     */
    function transferFrom(address _from, address _to, uint256 _value) public returns (bool success) {
        require(_value <= allowance[_from][msg.sender]);     // Check allowance
        allowance[_from][msg.sender] -= _value;
        _transfer(_from, _to, _value);
        return true;
    }

    /**
     * Set allowance for other address
     *
     * Allows `_spender` to spend no more than `_value` tokens in your behalf
     *
     * @param _spender The address authorized to spend
     * @param _value the max amount they can spend
     */
    function approve(address _spender, uint256 _value) public
        returns (bool success) {
        allowance[msg.sender][_spender] = _value;
        return true;
    }

    /**
     * Set allowance for other address and notify
     *
     * Allows `_spender` to spend no more than `_value` tokens in your behalf, and then ping the contract about it
     *
     * @param _spender The address authorized to spend
     * @param _value the max amount they can spend
     * @param _extraData some extra information to send to the approved contract
     */
    function approveAndCall(address _spender, uint256 _value, bytes _extraData)
        public
        returns (bool success) {
        tokenRecipient spender = tokenRecipient(_spender);
        if (approve(_spender, _value)) {
            spender.receiveApproval(msg.sender, _value, this, _extraData);
            return true;
        }
    }

    /**
     * Destroy tokens
     *
     * Remove `_value` tokens from the system irreversibly
     *
     * @param _value the amount of money to burn
     */
    function burn(uint256 _value) public returns (bool success) {
        require(balanceOf[msg.sender] >= _value);   // Check if the sender has enough
        balanceOf[msg.sender] -= _value;            // Subtract from the sender
        totalSupply -= _value;                      // Updates totalSupply
        Burn(msg.sender, _value);
        return true;
    }

    /**
     * Destroy tokens from other account
     *
     * Remove `_value` tokens from the system irreversibly on behalf of `_from`.
     *
     * @param _from the address of the sender
     * @param _value the amount of money to burn
     */
    function burnFrom(address _from, uint256 _value) public returns (bool success) {
        require(balanceOf[_from] >= _value);                // Check if the targeted balance is enough
        require(_value <= allowance[_from][msg.sender]);    // Check allowance
        balanceOf[_from] -= _value;                         // Subtract from the targeted balance
        allowance[_from][msg.sender] -= _value;             // Subtract from the sender's allowance
        totalSupply -= _value;                              // Update totalSupply
        Burn(_from, _value);
        return true;
    }
}

contract owned {
    address public owner;

    function owned() public {
        owner = msg.sender;
    }

    modifier onlyOwner {
        require(msg.sender == owner);
        _;
    }

    function transferOwnership(address newOwner) public onlyOwner {
        owner = newOwner;
    }
}

contract SurveyToken is TokenERC20, owned
{
    struct Survey {
        address initiator;
        uint256 toPay;
        uint256 balance;
        uint32 tickets;
        uint256 reward;
        mapping(address => bool) respondents;
    }

    address feeReceiver;

    mapping(bytes32 => Survey) surveys;
    mapping(address => bool) robots;

    modifier onlyRobot {
        require(robots[msg.sender]);
        _;
    }

    function SurveyToken(uint256 initialSupply) public
    TokenERC20(initialSupply) {
        feeReceiver = msg.sender;
    }

    function setFeeReceiver(address newReceiver) public onlyOwner {
        require(newReceiver != 0x0);
        feeReceiver = newReceiver;
    }

    function addRobot(address newRobot) public onlyOwner returns(bool success) {
        require(newRobot != 0x0);
        require(robots[newRobot] == false);

        robots[newRobot] = true;
        return true;
    }
    function removeRobot(address oldRobot) public onlyOwner returns(bool success) {
        require(oldRobot != 0x0);
        require(robots[oldRobot] == true);

        robots[oldRobot] = false;
        return true;
    }

    function placeNewSurvey(bytes32 key, uint256 toPay, uint32 tickets, uint256 reward) public returns(bool success) {
        require(surveys[key].initiator == 0x0);
        require(tickets > 0 && reward >= 0);
        uint256 rewardBalance = tickets * reward;
        require(rewardBalance < toPay && toPay > 0);
        require(balanceOf[msg.sender] >= toPay);
        
        uint256 fee = toPay - rewardBalance;
        require(balanceOf[feeReceiver] + fee > balanceOf[feeReceiver]);
        transfer(feeReceiver, fee);
        
        balanceOf[msg.sender] -= rewardBalance;
        surveys[key] = Survey(msg.sender, toPay, rewardBalance, tickets, reward);
        Transfer(msg.sender, 0x0, rewardBalance);
        return true;
    }

    function giveReward(bytes32 surveyKey, address respondent, uint8 karma) public onlyRobot returns(bool success) {
        require(respondent != 0x0);
        Survey storage surv = surveys[surveyKey];
        require(surv.respondents[respondent] == false);
        require(surv.tickets > 0 && surv.reward > 0 && surv.balance >= surv.reward);
        require(karma >= 0 && karma <= 10);
        
        if (karma < 10) {
            uint256 fhalf = surv.reward / 2;
            uint256 shalf = ((surv.reward - fhalf) / 10) * karma;
            uint256 respReward = fhalf + shalf;
            uint256 fine = surv.reward - respReward;
            
            require(balanceOf[respondent] + respReward > balanceOf[respondent]);
            require(balanceOf[feeReceiver] + fine > balanceOf[feeReceiver]);
            
            balanceOf[respondent] += respReward;
            Transfer(0x0, respondent, respReward);
            
            balanceOf[feeReceiver] += fine;
            Transfer(0x0, feeReceiver, fine);
        } else {
            require(balanceOf[respondent] + surv.reward > balanceOf[respondent]);
            balanceOf[respondent] += surv.reward;
            Transfer(0x0, respondent, surv.reward);
        }

        surv.tickets--;
        surv.balance -= surv.reward;
        surv.respondents[respondent] = true;
        return true;
    }
    
    function removeSurvey(bytes32 surveyKey) public onlyRobot returns(bool success) {
        Survey storage surv = surveys[surveyKey];
        require(surv.initiator != 0x0 && surv.balance > 0);
        require(balanceOf[surv.initiator] + surv.balance > balanceOf[surv.initiator]);
        
        balanceOf[surv.initiator] += surv.balance;
        Transfer(0x0, surv.initiator, surv.balance);
        surv.balance = 0;
        return true;
    }

    function getSurveyInfo(bytes32 key) public constant returns(bool success, uint256 toPay, uint32 tickets, uint256 reward) {
        Survey storage surv = surveys[key];
        require(surv.initiator != 0x0);

        return (true, surv.toPay, surv.tickets, surv.reward);
    }
}